#-swank
(unless (member :child-sbcl *features*)
  (quit
   :unix-status
   (process-exit-code
    (run-program *runtime-pathname*
                 `("--control-stack-size" "128MB"
                   "--noinform" "--disable-ldb" "--lose-on-corruption" "--end-runtime-options"
                   "--eval" "(push :child-sbcl *features*)"
                   "--script" ,(namestring *load-pathname*))
                 :output t :error t :input t))))
(eval-when (:compile-toplevel :load-toplevel :execute)
  (sb-int:defconstant-eqx opt
    #+swank '(optimize (speed 3) (safety 2))
    #-swank '(optimize (speed 3) (safety 0) (debug 0))
    #'equal)
  #+swank (ql:quickload '(:cl-debug-print :fiveam) :silent t)
  #-swank (set-dispatch-macro-character
           #\# #\> (lambda (s c p) (declare (ignore c p)) `(values ,(read s nil nil t)))))
#+swank (cl-syntax:use-syntax cl-debug-print:debug-print-syntax)
#-swank (disable-debugger) ; for CS Academy

;; BEGIN_INSERTED_CONTENTS
(defmacro with-buffered-stdout (&body body)
  "Buffers all outputs to *STANDARD-OUTPUT* in BODY and flushes them to
*STANDARD-OUTPUT* after BODY has been done (without error). Note that only
BASE-CHAR is allowed."
  (let ((out (gensym)))
    `(let ((,out (make-string-output-stream :element-type 'base-char)))
       (let ((*standard-output* ,out))
         ,@body)
       (write-string (get-output-stream-string ,out)))))

(declaim (ftype (function * (values fixnum &optional)) read-fixnum))
(defun read-fixnum (&optional (in *standard-input*))
  "NOTE: cannot read -2^62"
  (declare #.OPT)
  (macrolet ((%read-byte ()
               `(the (unsigned-byte 8)
                     #+swank (char-code (read-char in nil #\Nul))
                     #-swank (sb-impl::ansi-stream-read-byte in nil #.(char-code #\Nul) nil))))
    (let* ((minus nil)
           (result (loop (let ((byte (%read-byte)))
                           (cond ((<= 48 byte 57)
                                  (return (- byte 48)))
                                 ((zerop byte) ; #\Nul
                                  (error "Read EOF or #\Nul."))
                                 ((= byte #.(char-code #\-))
                                  (setq minus t)))))))
      (declare ((integer 0 #.most-positive-fixnum) result))
      (loop
        (let* ((byte (%read-byte)))
          (if (<= 48 byte 57)
              (setq result (+ (- byte 48)
                              (* 10 (the (integer 0 #.(floor most-positive-fixnum 10))
                                         result))))
              (return (if minus (- result) result))))))))

(in-package :cl-user)

(defmacro dbg (&rest forms)
  #+swank
  (if (= (length forms) 1)
      `(format *error-output* "~A => ~A~%" ',(car forms) ,(car forms))
      `(format *error-output* "~A => ~A~%" ',forms `(,,@forms)))
  #-swank (declare (ignore forms)))

(defmacro define-int-types (&rest bits)
  `(progn
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "UINT~A" b)) () '(unsigned-byte ,b))) bits)
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "INT~A" b)) () '(signed-byte ,b))) bits)))
(define-int-types 2 4 7 8 15 16 31 32 62 63 64)

(declaim (inline println))
(defun println (obj &optional (stream *standard-output*))
  (let ((*read-default-float-format* 'double-float))
    (prog1 (princ obj stream) (terpri stream))))

(defconstant +mod+ 1000000007)

;;;
;;; Body
;;;

(defun main ()
  (declare #.OPT)
  (let* ((n (read))
         (res (make-array n :element-type 'uint62))
         (cs (make-array n :element-type 'uint31))
         (graph (make-array n :element-type 'list :initial-element nil))
         (dp (make-array n :element-type 'hash-table))
         (sizes (make-array n :element-type 'uint32)))
    (declare (uint31 n))
    (dotimes (i n)
      (setf (aref cs i) (- (read-fixnum) 1)))
    (dotimes (i (- n 1))
      (let ((a (- (read-fixnum) 1))
            (b (- (read-fixnum) 1)))
        (push a (aref graph b))
        (push b (aref graph a))))
    (sb-int:named-let dfs ((v 0) (parent -1))
      (let ((res 1))
        (declare (uint31 res))
        (dolist (child (aref graph v))
          (declare (uint31 child))
          (unless (= child parent)
            (dfs child v)
            (incf res (aref sizes child))))
        (setf (aref sizes v) res)))
    (sb-int:named-let dfs ((v 0) (parent -1))
      (let ((table (make-hash-table :test #'eq))
            (c (aref cs v)))
        (labels ((%merge (table1 table2)
                   (when (> (hash-table-count table2)
                            (hash-table-count table1))
                     (rotatef table1 table2))
                   (maphash (lambda (col delta)
                              (declare (uint31 col delta))
                              (multiple-value-bind (num present-p) (gethash col table1)
                                (if present-p
                                    (setf (gethash col table1) (+ (the uint31 num) delta))
                                    (setf (gethash col table1) delta))))
                            table2)
                   table1))
          (dolist (child (aref graph v))
            (declare (uint31 child))
            (unless (= child parent)
              (dfs child v)
              (let* ((child-table (aref dp child))
                     (card (- (aref sizes child)
                              (the uint31 (or (gethash c child-table) 0)))))
                (declare (uint31 card))
                (incf (aref res c) (ash (* card (+ card 1)) -1)))
              (setq table (%merge table (aref dp child)))))
          (setf (gethash c table) (aref sizes v))
          (setf (aref dp v) table))))
    (let ((root-table (aref dp 0)))
      (maphash (lambda (col num)
                 (declare (uint31 col num))
                 (let ((card (- n num)))
                   (declare (uint31 card))
                   (incf (aref res col) (ash (* card (+ card 1)) -1))))
               root-table)
      (dotimes (c n)
        (unless (gethash c root-table)
          (setf (aref res c) (ash (* n (+ n 1)) -1)))))
    (with-buffered-stdout
      (dotimes (i n)
        (let ((x (aref res i)))
          (println (- (ash (* n (+ n 1)) -1) x)))))))

#-swank (main)

;;;
;;; Test and benchmark
;;;

#+swank
(defun io-equal (in-string out-string &key (function #'main) (test #'equal))
  "Passes IN-STRING to *STANDARD-INPUT*, executes FUNCTION, and returns true if
the string output to *STANDARD-OUTPUT* is equal to OUT-STRING."
  (labels ((ensure-last-lf (s)
             (if (eql (uiop:last-char s) #\Linefeed)
                 s
                 (uiop:strcat s uiop:+lf+))))
    (funcall test
             (ensure-last-lf out-string)
             (with-output-to-string (out)
               (let ((*standard-output* out))
                 (with-input-from-string (*standard-input* (ensure-last-lf in-string))
                   (funcall function)))))))

#+swank
(defun get-clipbrd ()
  (with-output-to-string (out)
    (run-program "powershell.exe" '("-Command" "Get-Clipboard") :output out :search t)))

#+swank (defparameter *this-pathname* (uiop:current-lisp-file-pathname))
#+swank (defparameter *dat-pathname* (uiop:merge-pathnames* "test.dat" *this-pathname*))

#+swank
(defun run (&optional thing (out *standard-output*))
  "THING := null | string | symbol | pathname

null: run #'MAIN using the text on clipboard as input.
string: run #'MAIN using the string as input.
symbol: alias of FIVEAM:RUN!.
pathname: run #'MAIN using the text file as input."
  (let ((*standard-output* out))
    (etypecase thing
      (null
       (with-input-from-string (*standard-input* (delete #\Return (get-clipbrd)))
         (main)))
      (string
       (with-input-from-string (*standard-input* (delete #\Return thing))
         (main)))
      (symbol (5am:run! thing))
      (pathname
       (with-open-file (*standard-input* thing)
         (main))))))

#+swank
(defun gen-dat ()
  (uiop:with-output-file (out *dat-pathname* :if-exists :supersede)
    (format out "")))

#+swank
(defun bench (&optional (out (make-broadcast-stream)))
  (time (run *dat-pathname* out)))

;; To run: (5am:run! :sample)
#+swank
(it.bese.fiveam:test :sample
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "3
1 2 1
1 2
2 3
"
    "5
4
0
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "1
1
"
    "1
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "2
1 2
1 2
"
    "2
2
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "5
1 2 3 4 5
1 2
2 3
3 4
3 5
"
    "5
8
10
5
5
"))
  (it.bese.fiveam:is
   (common-lisp-user::io-equal "8
2 7 2 5 4 1 7 5
3 1
1 2
2 7
4 5
5 6
6 8
7 8
"
    "18
15
0
14
23
0
23
0
")))


Measured memory_bytes: 282044